3. A retired engineer has $250,000 to invest and is willing to spend five hours per week managing her investments. Municipal bonds earn 6% per year and require no management. Real estate investments are expected to appreciate at 8% per year and require one hour of management per $100,000 invested. Blue chip stocks earn 10% per year and require 1.5 hours of management. Junk bonds earn 12% and require 2.5 hours, while grain futures earn 15% and require five hours per $100,000 invested.

(i) How should the retiree invest her money in order to maximize her expected earnings? Solve as a linear programming problem.

The decision variables will be x_i with the meanings

$x_1 =$ amount invested in municipal bonds.
$x_2 =$ amount invested in real estate.
$x_3 =$ amount invested in blue-chip stocks.
$x_4 =$ amount invested in junk bonds.
$x_5 =$ amount invested in grain futures.

Let h_i be the hours of time needed per $100,000 invested and r_i be the return rates for each of the respective catagories. Thus,

$$h = (0, 1, 1.5, 2.5, 5) \qquad \text{and} \qquad r = (0.06, 0.08, 0.10, 0.12, 0.15).$$

The corresponding constraints are on the hours available to manage the investments

$$H(x) = h \cdot x/100000 \leq 5$$

and the total amount of money available

$$T(x) = \sum_{i=1}^{5} x_i \leq 250000.$$

The objective function is the earnings

$$R(x) = r \cdot x$$

In [1]:
```
using JuMP, HiGHS
```

In [2]:
```
model=Model(HiGHS.Optimizer)
```

Out[2]:
```
A JuMP Model
├ solver: HiGHS
├ objective_sense: FEASIBILITY_SENSE
├ num_variables: 0
├ num_constraints: 0
└ Names registered in the model: none
```

In [3]:
```
@variable(model,x[1:5].>=0)
```

Out[3]:
```
5-element Vector{VariableRef}:
 x[1]
 x[2]
 x[3]
 x[4]
 x[5]
```

In [4]:
```
h=[0,1,1.5,2.5,5]
```

Out[4]:
```
5-element Vector{Float64}:
 0.0
 1.0
 1.5
 2.5
 5.0
```

In [5]:
```
r=[0.06,0.08,0.10,0.12,0.15]
```

Out[5]:
```
5-element Vector{Float64}:
 0.06
 0.08
 0.1
 0.12
 0.15
```

In [6]:
```
H(x)=h'*x/100000
H(x)
```

Out[6]: $1.0 \times 10^{-5} x_2 + 1.5 \times 10^{-5} x_3 + 2.5 \times 10^{-5} x_4 + 5.0 \times 10^{-5} x_5$

In [7]:
```
T(x)=ones(5)'*x
T(x)
```

Out[7]: $x_1 + x_2 + x_3 + x_4 + x_5$

In [8]:
```
R(x)=r'*x
R(x)
```

Out[8]: $0.06x_1 + 0.08x_2 + 0.1x_3 + 0.12x_4 + 0.15x_5$

In [9]:
```
@objective(model,Max,R(x))
```

Out[9]: $0.06x_1 + 0.08x_2 + 0.1x_3 + 0.12x_4 + 0.15x_5$

In [10]:
```
c1=@constraint(model,H(x)<=5)
```

Out[10]:

$$1.0 \times 10^{-5} x_2 + 1.5 \times 10^{-5} x_3 + 2.5 \times 10^{-5} x_4 + 5.0 \times 10^{-5} x_5 \leq 5$$

In [11]:

```
c2=@constraint(model,T(x)<=250000)
```

Out[11]:

$$x_1 + x_2 + x_3 + x_4 + x_5 \leq 250000$$

In [12]:

```
print(model)
```

$$\begin{aligned} \max \quad & 0.06x_1 + 0.08x_2 + 0.1x_3 + 0.12x_4 + 0.15x_5 \\ \text{Subject to} \quad & 1.0 \times 10^{-5} x_2 + 1.5 \times 10^{-5} x_3 + 2.5 \times 10^{-5} x_4 + 5.0 \times 10^{-5} x_5 \leq 5 \\ & x_1 + x_2 + x_3 + x_4 + x_5 \leq 250000 \\ & x_1 \geq 0 \\ & x_2 \geq 0 \\ & x_3 \geq 0 \\ & x_4 \geq 0 \\ & x_5 \geq 0 \end{aligned}$$

In [13]:

```
optimize!(model)
```

```
Running HiGHS 1.13.1 (git hash: 1d267d97c): Copyright (c) 2026 under Apache
2.0 license terms
Using BLAS: blastrampoline
LP has 2 rows; 5 cols; 9 nonzeros
Coefficient ranges:
  Matrix  [1e-05, 1e+00]
  Cost    [6e-02, 1e-01]
  Bound   [0e+00, 0e+00]
  RHS     [5e+00, 2e+05]
Presolving model
2 rows, 5 cols, 9 nonzeros  0s
2 rows, 5 cols, 9 nonzeros  0s
Presolve reductions: rows 2(-0); columns 5(-0); nonzeros 9(-0) - Not reduced
Problem not reduced by presolve: solving the LP
Using dual simplex solver
  Iteration        Objective     Infeasibilities num(sum)
          0    -6.9599912765e+00 Ph1: 2(9.4152); Du: 5(6.95999) 0.0s
          3     2.7500000000e+04 Pr: 0(0) 0.0s

Model status        : Optimal
Simplex   iterations: 3
Objective value     :  2.7500000000e+04
P-D objective error :  1.3228773321e-16
HiGHS run time      :          0.00
```

In [14]:

```
objective_value(model)
```

Out[14]: 27500.000000000007

In [15]:

```
value(x)
```

Out[15]:
```
5-element Vector{Float64}:
      0.0
      0.0
 125000.00000000016
 124999.99999999993
      0.0
```

The retiree should invest $125000 in blue-chip stocks and $125000 in junk bonds.

The earnings will be $27500 per year.

(ii) Determine the shadow prices for each constraint. Interpret each shadow price in the context of this problem.

In [16]:
```
# shadow price for the H(x)<=5 constraint on time
dual(c1)
```

Out[16]:
```
-2000.0000000000045
```

This means if the available time to manage the investments were increased by 1 hour/week then earnings would increase by $2000 per year.

In [17]:
```
# shadow price for the T(x)<=250000 constraint on money
dual(c2)
```

Out[17]:
```
-0.06999999999999997
```

This means if the amount invested were increased by $1 the earnings would increase by $0.07 per year.

(iii) The retiree downloads software from the internet that allows her to effectively manage her grain futures in three hours per week per $100,000. How does this change the results in parts (i) and (ii)?

In [18]:
```
h=[0,1,1.5,2.5,3]
```

Out[18]:
```
5-element Vector{Float64}:
 0.0
 1.0
 1.5
 2.5
 3.0
```

In [19]:
```
delete(model,c1)
c1=@constraint(model,H(x)<=5)
```

Out[19]:

$$1.0 \times 10^{-5} x_2 + 1.5 \times 10^{-5} x_3 + 2.5 \times 10^{-5} x_4 + 3.0 \times 10^{-5} x_5 \leq 5$$

In [20]:
```
print(model)
```

$$
\begin{aligned}
\max \quad & 0.06x_1 + 0.08x_2 + 0.1x_3 + 0.12x_4 + 0.15x_5 \\
\text{Subject to} \quad & x_1 + x_2 + x_3 + x_4 + x_5 \leq 250000 \\
& 1.0 \times 10^{-5} x_2 + 1.5 \times 10^{-5} x_3 + 2.5 \times 10^{-5} x_4 + 3.0 \times 10^{-5} x_5 \leq 5 \\
& x_1 \geq 0 \\
& x_2 \geq 0 \\
& x_3 \geq 0 \\
& x_4 \geq 0 \\
& x_5 \geq 0
\end{aligned}
$$

In [21]:
```
optimize!(model)
```

```
LP has 2 rows; 5 cols; 9 nonzeros
Coefficient ranges:
  Matrix  [1e-05, 1e+00]
  Cost    [6e-02, 1e-01]
  Bound   [0e+00, 0e+00]
  RHS     [5e+00, 2e+05]
Solving LP with useful basis so presolve not used
Using dual simplex solver
  Iteration        Objective     Infeasibilities num(sum)
          0    -7.1999795117e-01 Ph1: 2(2.07344); Du: 2(0.719998) 0.0s
          2     3.0000000000e+04 Pr: 0(0) 0.0s

Model status        : Optimal
Simplex   iterations: 2
Objective value     :  3.0000000000e+04
P-D objective error :  6.0631969585e-17
HiGHS run time      :          0.00
```

In [22]:
```
objective_value(model)
```

Out[22]:
```
30000.000000000004
```

In [23]:
```
objective_value(model)-27500
```

Out[23]:
```
2500.0000000000036
```

In [24]:
```
value(x)
```

Out[24]:
```
5-element Vector{Float64}:
  83333.33333333333
      0.0
      0.0
      0.0
 166666.6666666667
```

The optimal investment strategy is now $83333.33 in municipal bonds and $166666.67 in grain futures.

The earnings have increased by $2500 and are now $30000 per/year.

In [25]:
```
dual(c1)
```

Out[25]:
```
-3000.0
```

In [26]:
```
dual(c2)
```

Out[26]:
```
-0.06
```

The contraint on time is even more costly so an additional hour/week would result in a $3000 increase in yearly earnings.

On the other hand, increasing the amount available to invest by $1 only increases the return by $0.06. Presumably, because the time constraint is so costly.

(iv) After a few disasters in the futures market, the engineer decides that risk is a significant factor in her investment strategy. An investment self-help book ranks municipal bonds, real estate, blue chip stocks, junk bonds, and grain futures as risk level 1, 4, 3, 6, and 10 respectively. The engineer decides that her investment portfolio should have an average risk level of no more than 4. How does this change the results in parts (i) and (ii)?

Let $v = (1, 4, 3, 6, 10)$ and introduct the new constraint

$$V(x) = v \cdot x / 250000 \leq 4.$$

In [27]:
```
v=[1,4,3,6,10]
V(x)=v'*x/250000
V(x)
```

Out[27]: $4.0 \times 10^{-6} x_1 + 1.6 \times 10^{-5} x_2 + 1.2 \times 10^{-5} x_3 + 2.4 \times 10^{-5} x_4 + 4.0 \times 10^{-5} x_5$

In [28]:
```
c3=@constraint(model,V(x)<=4)
```

Out[28]:

$$4.0 \times 10^{-6} x_1 + 1.6 \times 10^{-5} x_2 + 1.2 \times 10^{-5} x_3 + 2.4 \times 10^{-5} x_4 + 4.0 \times 10^{-5} x_5 \leq 4$$

In [29]:
```
optimize!(model)
```

```
LP has 3 rows; 5 cols; 14 nonzeros
Coefficient ranges:
  Matrix  [4e-06, 1e+00]
  Cost    [6e-02, 1e-01]
  Bound   [0e+00, 0e+00]
  RHS     [4e+00, 2e+05]
Solving LP with useful basis so presolve not used
Using dual simplex solver
  Iteration        Objective     Infeasibilities num(sum)
          0     2.9999949658e+04 Pr: 1(6144) 0.0s
          2     2.6785714286e+04 Pr: 0(0) 0.0s

Model status        : Optimal
Simplex   iterations: 2
Objective value     :  2.6785714286e+04
P-D objective error :  0.0000000000e+00
HiGHS run time      :          0.00
```

In [30]:
```
objective_value(model)
```

Out[30]:
```
26785.714285714283
```

In [31]:
```
value(x)
```

Out[31]:
```
5-element Vector{Float64}:
      0.0
      0.0
 214285.71428571423
      0.0
  35714.28571428573
```

To reduce risk level the new strategy invests $214285.71 in blue-chip stocks and $35714.28 in grain futures.

The yearly earnings on the investment is reduced to $26785.71 per year.

In [32]:
```
dual(c1)
```

Out[32]:
```
0.0
```

In [33]:
```
dual(c2)
```

Out[33]:
```
-0.07857142857142857
```

In [34]:
```
dual(c3)
```

Out[34]:
```
-1785.7142857142849
```

The fact that the shadow price for the time constraint is 0 implies that time needed to manage the investments is no longer a factor that limits the earnings.

Every increase of $1 in the investment results in a $0.07857 increase in yearly earnings.

Allowing 1 unit greater risk factor would increase yearly earnings by $1785.71. But note these are average yearly earnings and the primary reason for the risk factor was to reduce the risk of the earnings being less than expected.

In [ ]: